







\$13.95

COMPUTER PROGRAMMING FOR KIDS AND OTHER BEGINNERS

APPLE II Edition



BY Royal
Van Horn

COMPUTER PROGRAMMING for KIDS AND OTHER BEGINNERS

APPLE II Edition

by
Royal Van Horn
Department of Elementary
And Secondary Education
University of North Florida


STERLING
SWIFT

publishing company
7901 South IH-35
austin, texas 78744

**Copyright © 1982
Sterling Swift Publishing Company**

ISBN-0-88408-151-6

To Hilary, Michael, and Sharon for their love, encouragement, and the time I stole from them; to Dr. John Minahan for establishing an environment that makes projects like this possible; and to Sterling Swift for his faith in me and his conviction that computers are for people.


TABLE OF CONTENTS	page
How To Read This Book	1
Chapter 1—Good News And Bad News	3
Chapter 2—All About The Apple Computer	5
Chapter 3—Getting Started	9
Turning On The Apple	10
RETURN	12
SHIFT	13
HOME	14
NEW	14
Chapter 4—Programs.....	15
GR	16
COLOR	16
PLOT	17
RUN	17
TEXT	17
LIST	17
Typing Mistakes	18
	18
Adding Lines To A Program ..	19
Color Numbers	21

	page
Chapter 5—Block Addresses	27
TEXT:HOME:LIST	32
Chapter 6—Designs	37
Variables	43
Chapter 7—Lines	45
HLIN	45
VLIN	49
Chapter 8—Easy BASIC words	57
PRINT " "	57
Punctuation	58
SPEED	61
GOTO	66
PRINT	69
Math	70
Strings	74
Chapter 9—Do Loops	79
FOR____TO____ and NEXT ..	80
Chapter 10—Do Loops And Graphics	85

	page
STEP	89
Chapter 11—More BASIC Words	93
REM	93
RND ()	95
INPUT	98
IF___THEN___	101
READ and DATA	103
Appendices	105

How To Read This Book

1. Read this book with a friend. That way you can help each other.
2. The important words in this book are in capital letters. RUN is an example. Keep a notebook for these words. Or, underline them as you read. Try to remember the words in capital letters!
3. You might find a page that's hard to read. Read hard pages two or three times.




NOTES:

Good News And Bad News


First, the good news. The Apple computer can do "graphics" in color! That means it can draw pictures. Many computers cannot do this. You can "program" it to do graphics. This book will teach you how.

Like all computers, the Apple computer can do math. It can also remember things for you.



It could remember all your grades. It could remember your friends' telephone numbers. It could even play games with you. You can learn to make it do these things.

Now the bad news. Spelling is important. You must be very careful. Words must be spelled right. Punctuation is also important. Periods (.), commas (,), quotation marks (") and parentheses () are very important.
BE CAREFUL.



All About The Apple Computer

chapter 2

The computer has five parts.

1. The Keyboard This part looks like a typewriter. You type on it. Then you push the **RETURN** button. This sends your message to the CPU. The message is called "input."

2. The CPU

CPU stands for "central processing unit." The CPU is the brain of a computer. You cannot see it. It is inside the machine. The CPU takes your "input" and "processes" it. That means it uses it. The CPU can only understand certain words or commands. These commands are stored in ROM.



3. ROM

This stands for Read Only Memory. You cannot change ROM. ROM contains BASIC. BASIC is a computer language. You must use BASIC words to use the computer. You cannot use English words. The computer speaks BASIC, not English.

4. RAM


This stands for Random Access Memory. It is like a notebook inside the computer. You write on it when you type. You can erase it, too. BASIC words must be used to do this. **When you turn the computer off, everything in RAM is erased.**



5. T.V. Screen


The T.V. is for output. Your computer prints messages on the T.V. screen. You put things in using the keyboard. The computer puts them out using the T.V.

Your computer has other parts. You'll learn about these other parts later. Now you're ready to start learning to program!



NOTES:

Turning the Apple computer on is easy. **First**, turn the T.V. on. Let it warm up. **Second**, does your computer have a disk drive? A disk drive looks like this.



DRIVE 1




Typing

The blinking box on the T.V. is called a **cursor**. The cursor means it's your turn to type. The cursor also tells where the next letter will go. Now type something. Type anything. Try all the letters and numbers.

When you're done typing push the **RETURN** key. This sends what you typed to the computer. The computer should print:

SYNTAX ERROR?

This means the computer doesn't understand what you typed. The reason is simple. You didn't use a BASIC word. Remember, the computer only understands BASIC words. Soon you will learn some easy BASIC words. But first here are some hints.



Things to Remember

■ means it's your
turn to type.

RETURN sends
what you typed to
the CPU.

Hint One: The **SHIFT** key

Look at the keys. Find the things that are on top of some of the keys. To type these, you must **hold down** the **SHIFT** button. This is true for anything on top of the key. Practice typing the things on top of the keys. Take time to do this now. Don't worry if the computer prints SYNTAX ERROR.

Hint Two: Ø and O, 1 and L

The zero on this computer has a line through it. It looks like this Ø. Zeros and Oh's are different. Don't confuse them. Type a few of each. See how they look. 1 and L are also different. But, they look the same. Type a few of these also.



**Things
to Remember**

To type some things
you must use the
SHIFT key.

Zero (Ø) and Oh (O)
are different.

One (1) and L are
different.

HOME

Push **RETURN**. Then type HOME. Then push **RETURN** again. Do this now. HOME erases the T.V. screen. But, it does **not** erase memory. It is very handy though.

NEW

To erase the computer's memory, type NEW. Then push **RETURN**. Do this now. This erases old programs. Always start out with a clean memory.

Be Careful

Don't turn a computer off and on quickly. This can hurt a computer. Wait 10 seconds between off and on.



Things to Remember

Type HOME to erase the screen. HOME does **not** erase memory.


Type NEW to erase the computer's memory. Always start with a clean memory.

Computers don't know anything. You might even say they're dumb! But, they can learn. When you teach a computer, you program it. A program tells a computer what to do. It also tells it how to do it. You must tell computers everything. To do this you program.

A program is made up of lines. Each line tells the computer to do something. Lines in a BASIC program have line numbers. That way the computer knows what to do first.


BASIC program lines are numbered by fives and tens. This is done so lines can be added later. Between 5 and 10 there are four numbers. You could add four lines between 5 and 10.

You could add lines 6, 7, 8, and 9 here. }	5 HOME
	10 COLOR=4



Things to Remember

Number program
lines by fives
or tens.

Here is your first program. Type the first line and check spelling. Fix mistakes with the  key. Then push the **RETURN** button. **Do this for all four lines.** Then you can type RUN to see it work. This program will light 1 block on the T.V. Soon you'll light lots of blocks.

Type This

Check Spelling

Read This

Then Push

NEW

RETURN

NEW erases the computer's memory.

Ø5 GR

RETURN

GR tells the computer it will use the T.V. for graphics.

1Ø COLOR=1 **RETURN**

COLOR= tells the computer to use color 1, red.

(There's more on the next page.)



Things to Remember

The  key is used to fix mistakes.

RETURN sends what you type to the computer.


NEW erases memory.

GR tells the computer to use the T.V. for graphics.

COLOR= is used to tell the computer a color.

15 PLOT 20, 20 **RETURN**

PLOT lights a small block on the T.V.



Things to Remember

R U N

RUN tells the computer to do your program.

PLOT lights one block.

Did you get a red block on the T.V.? If so, then G-R-E-A-T! If not, don't worry. It's probably a typing mistake. We'll talk about fixing mistakes in a minute.

RUN tells the computer to do your program

Where did the program go? It's in RAM. To look at it you must do three things. **Do these now**

1. Type TEXT and push **RETURN**. This tells the computer you want to use the T.V. for words.
2. Type HOME and push **RETURN**. This clears the screen—remember?
3. Type LIST and push **RETURN**. This gets your program from RAM.

TEXT tells the computer to use the whole T.V. for words.

HOME clears the screen.

Now your programs should be on the T.V. If it isn't — repeat steps 1, 2, and 3.

LIST gets a program from RAM.

You've already learned these important BASIC words.

IMPORTANT BASIC WORDS

Special Key

What it does when you push it

RETURN

sends what you've typed to the computer (CPU).



lets you correct typing mistakes. (only works **before** you push **RETURN** .)

Commands

Type without a line number to...

HOME

erases the screen—does **NOT** erase a program.

NEW

erases the program in RAM.

LIST

lets you look at a program (gets it from RAM)

RUN

starts a program running

TEXT

lets you use the whole TV screen for words

Statements

Use with a line number in a program to...

10 GR

tells the computer to use the TV for graphics

15 COLOR =

selects a color from 0 to 15 (colors are on the next page)

20 PLOT __, __ lights 1 block on the T.V.

Put a copy of this page by the computer

APPLE II COLOR NUMBERS


10 COLOR=____

Put one of these
numbers here ↗

to use this color ↘

0	Black
1	Magenta (Red)
2	Dark Blue
3	Purple
4	Dark Green
5	Light Grey
6	Medium Blue
7	Light Blue
8	Brown
9	Orange
10	Dark Grey
11	Pink
12	Light Green
13	Yellow
14	Aqua
15	White

Put a copy of this page by the computer



Remember these things about programs.


ALL ABOUT PROGRAMS

1. BASIC programs have line numbers.
2. All program lines must be typed correctly.
If there's one error a program won't work.
3. You must push **RETURN** after each line.
4. You can add or erase program lines.
5. **Programs almost never work right the first time. You often have to fix them. This is called debugging.**
6. Type these lines to LIST a graphics program.
TEXT **RETURN**
HOME **RETURN**
LIST **RETURN**

Lesson One

Name _____


PLOT __, __ PRACTICE

Here are some graphics programs. Each program uses PLOT __, __ to light some blocks. Type in a line. Check the typing on each line. Use the  key to fix mistakes. Then push **RETURN**. Put a check in the box when done.

Type all the lines.

- ☐ TEXT
- ☐ HOME
- ☐ NEW
- ☐ Ø5 GR
- ☐ 1Ø COLOR=7
- ☐ 15 PLOT 19, 36
- ☐ 2Ø PLOT 2Ø, 35
- ☐ 25 PLOT 2Ø, 34
- ☐ 3Ø PLOT 21, 36
- ☐ RUN

Draw the figure over here. 



Is your last program still in the computer?

Type these lines to see.

TEXT RETURN

HOME RETURN

LIST RETURN

Is your program in the computer? If it is, do

not type the first two lines.

NEW

10 GR

35 COLOR=13

40 PLOT 20, 0

45 PLOT 20, 2


50 PLOT 20, 4

55 PLOT 18, 2

60 PLOT 22, 2

RUN

Draw the figure over here. →



One more program.

TEXT

HOME

NEW

Ø5 GR

1Ø COLOR=6

15 PLOT 19, 18

2Ø PLOT 21, 18

25 COLOR=9

3Ø PLOT 18, 2Ø

35 PLOT 19, 21


4Ø PLOT 2Ø, 21

45 PLOT 21, 21

5Ø PLOT 22, 2Ø


RUN

Draw the figure over here. ➔



NOTES

Look at the picture on the next page. There are a lot of small blocks. Count the number of small blocks across the screen. Do this now before you read on. You should get 40. Now, look at the top of the picture. Find the small numbers. These numbers go from 0 to 39. If you start counting with 0 you will get 39, too. Now, count the blocks from top to bottom. Do this before you read on. You will get 39 if you start with 0. Try it. These numbers are on the left of the picture. Find these little numbers now.



Things to Remember

There are 40 blocks
across →

There are 40 blocks
down ↓


These blocks are
numbered from 0
to 39.

BLOCK ADDRESSES

FIRST COUNT OVER →


THEN COUNT DOWN ↓

START




Each little block has an address. It is a two number address. The first number is how far over it is. The second number is how far down it is. These are the 2 numbers after PLOT.

Example 1: Block 10, 5 is **over** 10 and **down** 5



Find block 10, 5 on the chart now.

Example 2: Block 3, 7 is **over** 3 and **down** 7.



Find block 3, 7 on the chart now.



Things to Remember

A block address is

over →

then down ↓

Block Colors

COLOR=__ tells the computer what color you want. A color number goes in the blank. Any number 0 to 15 will work. See the chart on page 21. You can change color anytime you want. Here are some examples.

Example 1: (Don't type this. It's just an example.)

```
05 COLOR=1
```

```
10 PLOT 20, 20
```

Lights a block. Its address is over 20, down 20. It will be **color 1, red**.


Example 2: (Don't type this.)

```
05 COLOR=9
```

```
10 PLOT 5, 20
```

```
15 PLOT 5, 30
```

Lights two blocks. One is over 5, down 20. The other is over 5, down 30. **Both** will be color 9, orange.



Things to Remember

COLOR=__ is used to pick a color. Color numbers 0-15 can go in the blank.

Example 3 (Don't type this.)

05 COLOR=1

10 PLOT 10, 3

15 COLOR=13

20 PLOT 3, 35

Lights two blocks. One will be color 1, red. It will be over 10, down 3. The other block will be color 13, yellow. It will be over 3, down 35.

Think of the PLOT statement like this

PLOT Number of Number of
 blocks over blocks down ↓
 → , ↓



**Things
to Remember**

PLOT_,_ lights
a block.

HINT

To look at a graphics program you type this:

TEXT RETURN


HOME RETURN

LIST RETURN

There is an easier way!

TEXT:HOME:LIST RETURN

all on one line. Don't forget the **:** between each word.



BLOCK ADDRESSES

Lesson Two

Name _____

This lesson will help you answer four questions about block addresses.

Program 1: What happens when both numbers are small?

TEXT:HOME:NEW

05 GR


10 COLOR=1

15 PLOT 0, 0

20 PLOT 2, 2

RUN

Where are the blocks when both numbers are small? Put an X on the picture.



What color were the blocks? Answer _____.



Program 2: What happens when both numbers are large?

Type this program.

TEXT:HOME:NEW

05 GR


10 COLOR=3

15 PLOT 39, 39

20 PLOT 37, 37

RUN

Where are the blocks when both numbers are large? Put an X on the picture.



What color were the blocks? Answer _____.

Program 3: What happens when the first number gets larger?

Type this program.

TEXT:HOME:NEW

05 GR

10 COLOR=12

15 PLOT 13, 0

20 PLOT 26, 0

25 PLOT 39, 0

RUN

Which way does the first number move the blocks? (**Hint:** across or down?)

Answer _____

What color were these blocks? Answer _____

Program 4: What happens when the second number gets larger?

Type this program.

TEXT:HOME:NEW

Ø GR

1Ø COLOR=11

15 PLOT Ø, 13

2Ø PLOT Ø, 26

25 PLOT Ø, 39

RUN

Which way does the second number move the blocks? (**Hint:** down or across?)

Answer _____

What color are these blocks? Answer _____

Designs


You can write your own graphics program. This chapter will teach you how. Look at the designs on the next page. Do this now.

Just read this now. You will do it later. To make a graphics program you:

1. Shade in blocks to make a design. Use colored pencils or sharp crayons. Try to use only a few blocks. This has been done on the next page.
2. Find all the blocks of the same color. Find the address of each block. First count over, then count down. Write the address under the right color. Do it like this.

<u>Blue</u>	<u>Red</u>	<u>Pink</u>
12, 39	12, 37	12, 35
12, 38	12, 36	12, 34
	11, 37	13, 35
	11, 36	13, 34


3. Write out the program. Use COLOR and PLOT lines.
4. Type in the program.



38

START

THEN COUNT DOWN ↓



DESIGNS

Let's do one of the designs on the last page.
Try the man first. Write the address numbers
in the blank spaces. Also write in the color
numbers. Then type in the program.

TEXT:HOME:NEW

05 GR

10 COLOR= _

15 PLOT __, __

20 PLOT __, __

25 COLOR= _

30 PLOT __, __

35 PLOT __, __

40 PLOT __, __

45 PLOT __, __

50 COLOR= _

55 PLOT __, __

60 PLOT __, __

65 PLOT __, __

70 PLOT __, __

RUN

} blue jeans

} red shirt

} pink head

Lesson Three

Name _____



Now, do one of the other figures. Write the form below on your paper. Then fill in the lines.

Is your last program still in the computer? If it is you can add to it. You won't need these two lines.

TEXT: HOME

NEW

10 GR

75 COLOR= _

80 PLOT __, __

85 PLOT __, __

90 PLOT __, __

95 PLOT __, __

100 PLOT __, __

105 PLOT __, __

110 PLOT __, __

You can add a COLOR= _
line any place you want.

Add more PLOT lines if you
need them. Cross out any
lines you don't need.

YOUR OWN DESIGN

Look at the next page. Use colored pencils to shade in a design. Don't use very many blocks. Find the address of each block. Make a list of these addresses under their color. Then do this lesson. It will help you with your program.

Don't type this program yet.

TEXT:HOME:NEW

01 GR

05 COLOR= _

10 PLOT _ , _

15 PLOT _ , _

20 PLOT _ , _

25 PLOT _ , _

30 PLOT _ , _

35 PLOT _ , _

40 PLOT _ , _

① Write the block addresses here.

② Add COLOR= _ lines where you want to change color.

③ Add more PLOT lines if you need them. Cross out any lines you don't need.

④ Check over the program.

Type it. RUN the program.

⑤ Debug the program if you need to.

Lesson Four

Name _____




BLOCK ADDRESSES

FIRST COUNT OVER ➡

THEN COUNT DOWN ⬇

START



Variables

The same thing can be written many different ways. Here are four ways to write one math problem.

$$\begin{array}{rcl} + & & = 4 \\ + & & = 4 \\ \square + \square & = & 4 \\ \bigcirc + \bigcirc & = & 4 \end{array}$$

These are all the same easy problem. You just find 2 numbers that add up to 4.

The space, line, square, and circle all mean the same thing. They mean, "put a number here."




Most computer languages don't use these blanks, circles, or squares. They use letters. In computer language the problem is written

$$X + Y = 4$$

The X and Y mean the same thing as blanks. They mean X and Y will become numbers. PLOT __, __ is usually written PLOT X, Y. They both mean the same. X and Y are called variables. They are called this because they change. They are different numbers in different programs.

Experiment: Take time now to write a few of your own programs. Try to program a graphic that is

1. an alphabet letter
 2. a small make-believe monster
 3. a pine tree
- Look in the back of this book for some ideas.



Things to Remember

PLOT X, Y means
the same as
PLOT __, __


Left-Right Lines

It takes a lot of PLOT's to draw a line. To draw one across the TV takes 40 PLOT's. You'd have to light blocks 0 to 39. Luckily there's an easier way. You can use the word HLIN. HLIN draws lines from left to right. The H stands for horizontal. It works like this.



Things to Remember

HLIN __, __ AT __
draws a left-right
line.




Here are some examples. **Don't type these yet** — just study them.

Example 1

1Ø HLIN Ø, 39 AT Ø


This draws a line across the **top** of the T.V. It's down Ø from the top.



Example 2

1Ø HLIN Ø, 39 AT 2Ø


This draws a line across the middle of the T.V. It's down 2Ø from the top.



Example 3

10 HLIN 0, 19 AT 39


This draws a line half way across the T.V. It will be AT 39, the bottom of the T.V.



Example 4

10 HLIN 20, 25 AT 30

This draws a shorter line. It will be 30 blocks down.



HLIN makes it a lot easier to do graphics. Here's an example. Type and RUN this program. Don't forget **RETURN** after each line.

```
TEXT:HOME:NEW
Ø5 GR
1Ø COLOR=2
15 HLIN 15, 17 AT 1Ø
2Ø HLIN 15, 17 AT 11
25 HLIN 15, 17 AT 12
3Ø COLOR=1
35 HLIN 18, 25 AT 1Ø
4Ø HLIN 18, 25 AT 12
45 HLIN 15, 25 AT 14
5Ø HLIN 15, 25 AT 16
RUN
```

You can stand and salute if you'd like. Oops — we forgot the flag pole.

Don't erase the program yet.

Up-Down Lines

VLIN is used to draw up and down lines. V stands for vertical. Think of VLIN as

VLIN top start number , bottom stop number **AT** number of lines over




Here are two examples. **Don't type these.**

Example 1


10 VLIN 0, 39 AT 0

This draws an up-down line on the left. It looks like this.



Things to Remember


VLIN __, __ AT __
draws an up-down
line.



Example 2

```
10 VLIN 20, 39 AT 30
```

This draws a shorter line. It will be over 30 from the left.



Now let's add the flag pole. Type this line to list your program.

```
TEXT:HOME:LIST
```

Now add these lines to your program.

```
55 COLOR=8
```

```
60 VLIN 10, 28 AT 13
```

Now RUN the program.

HLIN and VLIN

Type and RUN this program. It uses lines to make a graphic.

```
TEXT:HOME:NEW
```

```
Ø5 GR
```

```
1Ø COLOR=6
```

```
15 HLIN Ø, 39 AT 39
```

```
2Ø HLIN Ø, 39 AT 38
```

```
25 COLOR=8
```

```
3Ø HLIN 12, 15 AT 37
```

```
35 HLIN 11, 15 AT 36
```

```
4Ø HLIN 15, 16 AT 35
```

```
45 HLIN 15, 16 AT 34
```


```
5Ø COLOR=9
```

```
55 PLOT 17, 35
```

Draw the figure below. Can you tell what it's supposed to be? (**Hint: quack-quack**)

Lesson Five

Name _____



One more program.

```
TEXT:HOME:NEW
```

```
Ø5 GR
```

```
1Ø COLOR=4
```


```
15 HLIN Ø, 39 AT Ø
```

```
2Ø VLIN Ø, 39 AT 39
```

```
25 HLIN Ø, 39 AT 39
```

```
3Ø VLIN Ø, 39 AT Ø
```

Draw what happened over here. →



It's your turn. Draw a graphic on the next page. **Don't use too many lines or blocks.** Then program your design below. Put all the lines of the same color together. This will save typing.

Ø5 GR

1Ø COLOR=

15

2Ø

25

3Ø

35


Add as many lines
as you need.

BLOCK ADDRESSES

FIRST COUNT OVER →


THEN COUNT DOWN ↓

START



Experiment: Doing what is in this book isn't programming. It's reading and studying.
Take time now to write a few programs Here are some ideas.

1. Look at the graphic pictures in the back of this book.
2. Design your own picture.
3. Try to do one for a holiday like:
 - Christmas
 - Columbus Day
 - Easter
 - Fourth of July
 - Halloween
 - Thanksgiving
4. Program the picture.



NOTES:

Easy Basic Words

Output

Enough graphics for now. We'll do more graphics later. PLOT, VLIN, and HLIN are "output" commands. They put things on the T.V. PRINT is another "output" command. It is used to "output" words and numbers.

PRINT " "

See the " " marks? These are very important. They tell the computer, "don't mess with this stuff. Just put it on the screen." You can put anything you want inside these marks. Here is an example. Type and RUN this program.

```
TEXT:HOME:NEW
```


```
Ø5 HOME
```

```
1Ø PRINT "*****"
```

```
15 PRINT "put your name here"
```

```
2Ø PRINT "*****"
```

```
RUN
```



Things to Remember

PRINT " " puts
what's between the
marks on the
screen.

HOME can also be
used in a program
line.

Let's try one more PRINT " " program. This time, you decide what goes inside the " " marks.

NEW:HOME

Ø5 HOME

1Ø PRINT " put anything "

15 PRINT " you want "


2Ø PRINT " here "

25 PRINT " "

RUN

Punctuation

You can put several things on one PRINT line. But, what goes between them is important. A **comma** (,) or a **semi-colon** (;) may be used. A ; makes the output close together. A , makes the output in " **columns** ." There are 3 "columns" across the T.V.




**Things
to Remember**

; keeps the output
close together

, puts the output into
columns

They look like this.



Type and RUN this program. See the ; and , ?
They make a difference.

```
HOME:NEW
```

```
Ø5 HOME
```

```
1Ø PRINT "ONE" ; "TWO" ; "THREE"
```

```
15 PRINT "ONE", "TWO", "THREE"
```

```
RUN
```

Don't forget. **A ;** keeps things together. **A ,**
puts them into columns.

Skipping Lines

You can skip lines when you print. It's easy, just type PRINT. Don't put anything after it. Type and RUN this program. It shows how this works.

```
HOME:NEW
```

```
10 PRINT "HELLO", "THERE "
```

```
15 PRINT
```

```
20 PRINT "HOW ARE YOU? "
```

```
25 PRINT
```

```
30 PRINT
```

```
35 PRINT "I AM FINE "
```

```
RUN
```

Don't erase the program yet. Read on.



Things to Remember

PRINT followed by
nothing outputs a
blank line.


Slowing Down Programs

Computers are very fast. So fast it's hard to see what they do. Some programs are neater if they run slower. To slow down the APPLE type

`SPEED=100`

without a line number.

Do this now. The number after speed can be from 0 to 255. Now RUN your last program again.



Things to Remember

`SPEED= ____` slows down or speeds up the computer. Use a number from 0 to 255. 0 is slow, 255 is fast.

Try these other SPEED numbers. Then RUN your program again.

SPEED=0

SPEED=150

SPEED=255

SPEED=255 returns the APPLE to normal.

Now you should begin to use this command. Just type it whenever you want. It can help you see how a program works. But, it won't work with graphic programs.

Experiment

Write a program that PRINTS a message. Type SPEED=___ to slow down the program.

Drawing Pictures With PRINT

You can even draw pictures with PRINT lines! Here is an example. Type and RUN this program.

```
HOME:NEW
```

```
Ø5 HOME
```

```
1Ø PRINT " !!!!!! "
```

```
15 PRINT "!!!!!!!!!! "
```

```
2Ø PRINT " * * "
```

```
25 PRINT " L "
```

```
3Ø PRINT " XXX "
```

```
RUN
```

Here is another example. Type and RUN this program.

```
HOME:NEW
05 HOME
10 PRINT "BOOM"
15 PRINT "B-O-O-M"
20 PRINT "B--O--O--M"
25 PRINT "B---O---O---M"
30 PRINT "B----O----O----M"
35 PRINT "B-----O-----O-----M"
RUN
```

Type SPEED=100. Then
RUN the program again.

Fancy PRINT

Design your own PRINT program. Try to make a fancy sign. Or try to draw a picture. Use **SPEED**= to add interest. Use a lot of special keys like: * ! < > & ? . You could even write a poem or story. Do what ever you want. If your computer has a printer, ask someone how to use it now.

NEW

05 HOME

10 PRINT " Put "

15 PRINT " anything "

20 PRINT " you want "

25 PRINT " here. "

← Add more lines here if you need to.

RUN

Write what's on the T.V. over here. →

Lesson Six

Name _____



GOTO

GOTO tells the computer which line to do next. It tells it where to go in a program. Here is an example:

```
99 GOTO 5
```

It works like this. The computer does the lines from 5 to 98. Then it comes to 99. It sees the instruction GOTO 5. So it goes back to line 5. This makes the computer start all over again. It does this forever or until you push **RESET**. Computer people say, "It's stuck in a loop."

Try it! Type and RUN this program.


```
NEW:HOME
```

```
05 PRINT " Your first name "
```

```
10 PRINT " Your last name "
```

```
99 GOTO 5
```

```
RUN
```




↪ The program goes around this circle forever. Push **RESET** to stop the program.



Things to Remember

GOTO tells the computer which line to do next.

Press **RESET** or **CTRL RESET** if a program is "stuck in a loop."



On some APPLES you'll have to:

1. Hold down the **CTRL** key
2. And press the **RESET** key.

Here is another example of GOTO.

```
HOME:NEW
```

```
Ø5 HOME
```

```
1Ø PRINT "***** "
```

```
15 PRINT "+++++++" "
```


```
2Ø PRINT "!!!!!!!!!!!!!" "
```

```
99 GOTO 1Ø
```

Push RESET to stop the program. Don't erase it yet. Change Line 99 to

```
99 GOTO 2Ø
```

RUN the program. See the difference? Push RESET to stop the program. Try different SPEED= __ numbers to see how it works.



Things to Remember

You can GOTO any line. But there must be a line there.

PRINT a Picture and GOTO

Lesson Seven

Name _____

You can use PRINT " " lines to make a picture. Let's try it.

1. **Finish the program below.** Make a picture that looks like a rocket. Do this inside lines 10 to 25.
2. Make the rocket fly off the screen. To do this PRINT a lot of "!!!" marks under it. Use GOTO to do this.

NEW

05 HOME

10 PRINT " "

15 PRINT " "

20 PRINT " "

25 PRINT " "

30 PRINT " ! " "

35 PRINT " !!! " "

40 PRINT " !!!!! " "

50 GOTO ____

Explain what happened below.



Memory Boxes and Numbers

The computer has a **memory**. It is called **RAM**. **RAM** is like a lot of little boxes. To use memory you

1. Give a box a name
2. Tell what goes in the box

Here are some examples.

05 A=10

10 B=5

20 M=20

This looks like this inside a computer.

Memory Boxes

A	B	M
10	5	20

PRINT

PRINT is used to get numbers from memory. You just don't use the " " marks. PRINT A gets the number in box A. It puts it on the TV screen. PRINT A, B, M gets a number from each box. It prints them all on the T.V.



Things to Remember

A memory box must be given a name. This is usually one letter.

Numbers can be stored in memory boxes.


PRINT A outputs the number in Box A to the T.V.

MATH

You can do math when you PRINT.
 PRINT A+B will get both numbers. It adds them and puts the answer on the T.V.

The computer uses 2 special symbols to do math. Study the table below. Find the * and the /.

Math			
Problem	Example	Symbol	Type
Add	A+B	+	PRINT A+B
Subtract	A-B	-	PRINT A-B
Multiply	AxB	*	PRINT A*B
Divide	A÷B	/	PRINT A/B



Things to Remember

You can do math with PRINT lines.

* is used to multiply

/ is used to divide.

Memory, PRINT, and Math

Lesson Eight

Name _____

These programs put numbers into memory boxes. They PRINT numbers from the boxes. They also do math with the numbers.

HOME:NEW

05 HOME

10 A = 2

15 B = 78

20 C = 104

} Puts numbers into
memory boxes.

25 PRINT "THE NUMBERS ARE..."

30 PRINT A, B, C ← PRINTS them out.

35 PRINT "A + B + C = "

40 PRINT A + B + C

← Adds them and
PRINTS the
answer.

Answer A + B + C = _____



The computer can do very hard problems.
Here is an example:

HOME:NEW

Ø5 HOME

1Ø A=8

15 B=832

2Ø PRINT "8 PLUS 832 IS "

25 PRINT A+B

3Ø PRINT "832 TIMES 832 IS "

35 PRINT B*B

4Ø PRINT "832 DIVIDED BY 8 IS "

45 PRINT B/A

RUN

Write the answers below.

8 PLUS 832 IS _____

832 TIMES 832 IS _____

832 DIVIDED BY 8 IS _____

YOUR TURN! Write your own program.

NEW

Ø5 HOME

1Ø
15
2Ø

} Put some numbers
into memory boxes.

25 PRINT
3Ø PRINT

} PRINT them out.

35 PRINT
4Ø PRINT
45 PRINT

} Do some math if
you'd like.

5Ø RUN

Write your problems and answers below.


Problems

Answers

Strings

Now for some super neat stuff. You can put words or symbols into memory boxes. In computer language, words are called **strings**. You have to remember two things.

- Memory boxes for strings must have a \$ after them. Examples are A\$, B\$, X\$, and so on.
- The words must be in " " marks. Examples are "BOB" "FRED" "XXXX" and "+++++" "



Things to Remember

Words inside a computer are called **strings**.

String memory boxes must have a \$ in their name. Strings must be in quotes " ".

Example 1 (Don't type this yet)

```
10 A$="*****"
```

This puts five *'s into a memory box.

Example 2 (Don't type this yet)

```
15 B$="+++++"
```

This puts five plus signs into a box.

Example 3 (Don't type this yet)

```
20 PRINT A$
```

This prints ***** on the screen.

Example 4 (Don't type this yet)


```
25 PRINT A$+B$+A$
```

This prints

```
*****+++++*****
```

on the screen.

Surprise! You can add strings.



Things to Remember

You can add strings
together.

76

Name _____





Now, change the last program. Remember the GOTO? Type the line below. Then RUN the program.

```
50 GOTO 25
```

Push **RESET** to stop the program.

Draw what's on the screen below.



Here's another program.

```
HOME:NEW
```

```
Ø5 HOME
```

```
1Ø A$=" * * * * * * * * * * "
```

```
15 B$=" + + + + + + + + + + "
```

```
2Ø C$=" Ø Ø Ø Ø Ø Ø Ø Ø Ø Ø "
```

```
25 PRINT A$
```

```
3Ø PRINT A$+B$
```

```
35 PRINT A$+B$+C$
```


Now add some PRINT lines to the program.
Make the line numbers between 35 and 99.
Then, try

```
99 GOTO ____
```

Experiment! Then write your neatest program below.

Do Loops

A do loop tells the computer to do one thing many times. It's like getting a computer to go around in a circle. Every time it goes around once it does something. This is like you running around your house many times. Everytime you passed the front door you would count a number. The first time past the door you would say 1. The second time past the door you would say 2. The third time past the door you would say 3. And so on. This is a neat trick to get a computer to do. On the next page you'll see how to do it.



FOR, TO, and NEXT are BASIC words. They are used to make a do loop. FOR and TO must be on the same line. They start the computer counting. You must tell the computer what number to start with. You also must tell the computer how far to count. FOR X=1 TO 10 makes the computer count from 1 to 10. FOR C=5 TO 25 makes the computer count from 5 to 25. NEXT is used to tell it when to add one. Here is a program with a do loop. Study it carefully.



Things to Remember

Do Loops look like
this.

FOR X= _ TO _



NEXT X

TYPE THIS

READ THIS

NEW

05 HOME

10 FOR X=1 TO 10

This tells the computer it will count from 1 to 10.

15 PRINT X

This tells the computer to put the number on the T.V.


20 NEXT X

This tells the computer to go to line 10, add 1 to the count, and go down the program again.

RUN

Don't blink! Computers are very fast.

The computer goes around this circle 10 times.



Don't erase the last program. If you did, retype it. Change line 10 to any of the following lines. This will help you understand do loops. Try each line 10.

10 FOR X=5 TO 20

10 FOR X=25 TO 40

10 FOR X=100 TO 115

10 FOR X=-10 TO 0

Experiment. Make up your own line 10.

DO LOOP PRACTICE

Lesson Ten

Name _____

Type and RUN this program.

```
HOME:NEW
```

```
Ø5 HOME
```

```
1Ø FOR X=1 TO 12
```

```
15 PRINT "*****"
```

```
2Ø NEXT X
```

```
RUN
```

What happened? Answer _____

The next program does some math. Be careful, it is a hard program to type. Put a space before and after PLUS and IS in line 15.

```
HOME:NEW
```

```
Ø5 HOME
```

```
1Ø FOR X=3 TO 12
```

```
15 PRINT X; " PLUS "; X; " IS "; X+X
```

```
2Ø NEXT X
```

```
25 RUN
```

You might have to debug the last program.

What happened? Answer _____



**Be careful. This
is a hard line to
type.**

Now for a race. You will race the computer.
You will both count to 100. Type this program.

```
NEW
05 HOME
10 FOR X=1 TO 100
15 PRINT X
20 NEXT X
25 PRINT "THE COMPUTER IS DONE."
RUN
```

Push **RETURN** and start counting!

Who won? Answer _____

Experiment!

1. Change the last program. Fix it so you can win.
2. See if you can write your own program that has a do loop.

Do Loops And Graphics

More Graphics

Suppose you wanted the whole T.V. one color. You could use the HLIN statement. But there are 40 lines on the T.V. (0 to 39). So, you'd have to type a 40 line program. The first part would look like this. **DON'T TYPE THIS!**

```
05 GR
10 COLOR= __
15 HLIN 0, 39 AT 0
20 HLIN 0, 39 AT 1
25 HLIN 0, 39 AT 2
30 HLIN 0, 39 AT 3
35 HLIN 0, 39 AT 4
40 HLIN 0, 39 AT 5
45 HLIN 0, 39 AT 6
50 HLIN 0, 39 AT 7
55 HLIN 0, 39 AT 8
60 HLIN 0, 39 AT 9
```

After this you'd still have 30 lines to type! There's an easier way. The computer can count from 0 to 39.



Study this program carefully. **The C stands for count.**

TYPE THIS

READ THIS

HOME:NEW

Ø5 GR

1Ø COLOR=7

15 FOR C=Ø TO 39 The computer will count from
Ø to 39. C stands for count.

2Ø HLIN Ø, 39 AT C Each time through the loop a
line is drawn. The first time it
will be AT Ø. The second time it
will be AT 1. And so on.

25 NEXT C Makes the computer go to line
15. 1 is added to the count. Then
it goes down the program
again.

RUN

It goes around this loop 4Ø times. But, it is very fast.
So, it looks like 1 step. Really it's 4Ø steps.

The computer only did one thing in the do loop. Why not have it do two things? Study this program carefully.

TYPE THIS

READ THIS

TEXT:HOME:NEW

Remember this?

Ø5 GR

1Ø FOR C=Ø TO 15

Start of the do loop.

15 COLOR=C

Computer counts the color numbers. The first time it will be color Ø, Black. The second time it will be color 1, Red. And so on.

2Ø VLIN Ø, 39 AT C



Computer draws a line at Ø, then 1, and so on.

25 NEXT C

Makes the computer go to 1Ø. It adds 1 to the count. Then it goes down the program again.

RUN

It goes around this loop 16 (Ø to 15) times.



One more program. This one has something new in it!

TYPE THIS

READ THIS

TEXT:HOME:NEW

Ø5 GR

1Ø COLOR=3

15 FOR C=1 TO 39

2Ø HLIN Ø, C AT C

25 NEXT C

Starts the loop.

Computer draws a line from Ø to the count. It will be from Ø to 1. Then from Ø to 2. And, so on.

Closes the loop.

PLOT can also be used inside a do loop. Here is an example. Type and RUN this program.

TYPE THIS

READ THIS

TEXT:HOME:NEW

Ø5 GR

1Ø COLOR=1

15 FOR C=Ø TO 39

2Ø PLOT C, C

25 NEXT C

Starts the loop.

Lights block Ø, Ø, then 1, 1, then 2, 2. And so on.

Closes the loop.

STEP

The computer can count by two; 0, 2, 4, 6, 8, 10 and so on. It can also count by 5; 0, 5, 10, 15, 20 and so on. It can count by any number. STEP is used to do this. STEP is used right after FOR __ TO __. The next program uses STEP. The computer will count by 2. Type and RUN the program.

```
TEXT:HOME:NEW
05 GR
10 COLOR=14
15 FOR C=0 TO 39 STEP 2
20 HLIN 0, 39 AT C
25 NEXT C
```

Experiment with the last program. Change line 15. Try

STEP 3

STEP 4

STEP 10



Things to Remember

STEP __ makes the computer count by the number after it. It can only be used in a FOR __ TO __ line.

DO LOOPS AND GRAPHICS

Lesson Eleven

Name _____

Finish the next program. It should make the top half of the T.V. red. It should make the bottom half green.

```
05 _____  
10 COLOR=_____  
15 FOR C= ____ TO ____  
20 HLIN 0, 39 AT ____  
30 NEXT C  
35 COLOR=_____  
40 FOR C= ____ TO ____  
45 HLIN 0, 39 AT ____  
50 NEXT ____
```

RUN the program. Debug it if you need to.



Here is a program that uses STEP. Experiment with it. Try different STEP numbers. Also, change lines 20 and 30. Try shorter lines.

TEXT:HOME:NEW

05 GR

10 FOR C=0 TO 39 STEP 2

15 COLOR=1

20 VLIN 0, 39 AT C

25 COLOR=5

30 HLIN 0, 39 AT C

40 NEXT C

Write your neatest result below.

NOTES:

More Basic Words

This chapter is full of new BASIC words. They will help you write more interesting programs.

REM

REM is short for remark. REM lines are ignored by the computer. Good programmers use a lot of REM lines. REM lines help people remember how a program works. Type and RUN the program below.

```
TEXT: HOME:NEW
```

```
Ø1 REM PROGRAM TO DRAW A PYRAMID
```

```
Ø5 GR
```

```
1Ø COLOR=12
```

```
15 FOR C=2Ø TO 39
```

```
2Ø REM RIGHT SIDE
```

```
25 PLOT C, C
```


```
3Ø REM LEFT SIDE
```

```
35 PLOT 39 - C, C
```

```
4Ø REM CLOSE DO LOOP
```

```
45 NEXT C
```

```
RUN
```



Things to Remember

REM lines are ignored by the computer. Use them to explain your program.

Now, take out lines 01, 20, 30, and 40. Then TEXT:HOME:LIST the program. Now, RUN it again. It still works, but it's harder to read. YOU should start using REM lines in your programs. Use them to make your programs easier to understand.

RND ()

RND is short for random. Random means without pattern. Random numbers are not in any order. You can get the computer to pick such numbers. You just type this line.

```
15 R = INT(10*RND(1)) + 1
```

This line makes the computer pick a number. See the 10 in the line? The number will be between 1 and 10. It will put this number in memory box R. You could put 100 where the 10 is. Then the computer would pick a number from 1 to 100.

RUN the next two programs. Study the output to see how this line works.

Program 1

```
HOME:NEW
```

```
05 REM PRINT 15 RANDOM NUMBERS
```

```
10 FOR C=1 TO 15
```

```
15 R = INT(10*RND(1)) + 1
```

```
20 PRINT R
```

```
25 NEXT C
```



Things to Remember

$\text{INT}(10 \times \text{RND}(1)) + 1$
is used to pick a
number. The "10" can
be any number.

PROGRAM 2

HOME:NEW

05 REM PLOT BLOCKS IN A ROW

10 HOME

15 GR

20 COLOR=1

25 FOR C=1 TO 100

30 R = INT(39*RND(1)) + 1

35 PLOT R, 20

40 NEXT C

The RND(1) line can be used more than once. The next program uses it 3 times. It is used to pick a color number. It is used twice to pick a block address. Study this program carefully.

TYPE THIS**READ THIS**

TEXT:HOME:NEW

05 REM KALEIDOSCOPE

10 HOME

15 GR

20 C=INT(16*RND(1)) Picks a color from 0 to 15

25 X=INT(40*RND(1)) Picks a number from 0 to 39

30 Y=INT(40*RND(1)) Picks a number from 0 to 39

40 COLOR=C Sets the color

45 PLOT X, Y Lights a block

50 GOTO 20 Starts over at line 20

RUN

Push **RESET** to stop the program.

INPUT

INPUT does two things. It prints on the screen. And, it stops a program to get input. Here's an example.

```
10 INPUT "HOW FAR DOWN (0-39) ";Y
```

Prints HOW FAR DOWN (0-39) ? on the screen and stops. The program will start again when you type a number. The next program shows how it works.

```
TEXT:HOME:NEW
```

```
05 REM LINE DRAWING PROGRAM
```

```
10 TEXT
```

```
15 HOME
```

```
20 INPUT "HOW FAR DOWN (0-39) ";Y
```

```
25 GR
```

```
30 COLOR=7
```

```
35 HLIN 0, 39 AT Y
```

```
40 FOR C = 1 TO 1000
```


```
45 NEXT C
```

```
50 GOTO 10
```

```
RUN
```

Push RESET to stop the program.

These two lines make
the computer count to
1000 before it goes to
line 10.



Things to Remember

INPUT stops a program. The program will start after words or numbers are typed.

INPUT can also be used with strings. You just have to use the \$ sign. Here's an example. Type and RUN this program.

```
TEXT:HOME:NEW
```

```
05 REM PUT NAME IN A SENTENCE
```

```
10 HOME
```

```
15 INPUT "WHAT'S YOUR NAME "; N$
```

```
20 PRINT
```

```
25 PRINT N$; " IS A NICE NAME "
```

INPUT can also get several numbers at once. You just separate the numbers with commas. This is a very powerful idea. Try the next program and see.



TEXT:HOME:NEW

05 REM PLOT PRACTICE

10 TEXT

15 HOME

20 INPUT "TYPE A BLOCK ADDRESS X, Y "; X, Y

25 GR

30 COLOR=1

40 PLOT X, Y

45 FOR C=1 TO 1000

50 NEXT C


55 GOTO 10

Experiment


1. Write a program using INPUT that draws up and down lines.
2. Design a program to randomly draw lines on the screen.

IF __ THEN __

IF __ THEN __ statements are used in programs to make decisions. Look at the diagram below.



A test is placed between IF and THEN. For example, $G=N$. If G **is** equal N, the computer will do what comes after THEN. If G is **not** equal N, the computer will just go to the next line. Lots of things can be placed after THEN. Examples include: GOTO, END, and PRINT statements. The next program uses an IF __ THEN __ line. Study the program carefully. RUN it several times.



Things to Remember

IF (test) THEN __
is used to make decisions.

```
TEXT:HOME:NEW
05 REM GUESSING GAME
10 TEXT
15 HOME
20 N = INT(10*RND(1)) + 1
25 PRINT "GUESS MY NUMBER FROM 1-10"
30 INPUT G
35 IF G=N THEN GOTO 50
40 PRINT "NOPE. TRY AGAIN."
45 GOTO 30
50 PRINT "GOOD JOB "
```

IF __ THEN __ statements also work with strings. It just takes a different test. An example is `X$="YES"`. Add these lines to the last program. RUN it several times.

```
55 INPUT "AGAIN (YES/NO) "; A$
60 IF A$="YES" THEN GOTO 15
```

Experiment

1. Design a true-false test on the computer.
(**Hint:** use `IF A$="TRUE" THEN C=C+1` lines to count the number right.)
2. Design a program that lights blocks. Test each input to be sure it's not too large.


READ & DATA

Remember PLOT graphics? They took a lot of typing. Luckily, there's an easier way. Using READ & DATA is one easier way. As before, you'll need an address. You put these numbers into DATA lines. They have this form:

DATA X, Y, X, Y, X, Y, X, Y, X, Y, X, Y,

DATA X, Y, X, Y, X, Y, X, Y, X, Y, X, Y and so on.

Then you put a READ X, Y line inside a do loop. Here's a sample program.



Things to Remember

READ & DATA must both be used in a program. They allow for easy entry of many numbers or letters.

```
TEXT:HOME:NEW
Ø5 REM READ & DATA DEMO
1Ø GR
15 COLOR=9
2Ø READ X, Y
3Ø IF X = -1 THEN 99
4Ø PLOT X, Y
5Ø GOTO 2Ø
6Ø DATA Ø, Ø, 39, Ø, 39, 39
7Ø DATA Ø, 39, 2Ø, 2Ø, -1, -1
99 END
RUN
```

Experiment

1. You can add all the DATA lines you want.
Remember to put -1, -1 in the last DATA line. This tells line 3Ø (IF __ THEN __ line) when you're done.
2. Design a program with READ & DATA to do some math.

APPENDICES

A. Sample Programs	106
B. Saving Programs on Cassette Tape	113
C. Saving Programs on Disks	117
D. Graphics Screen Locations	120
E. Suggestions for Teachers and Parents ...	121
F. Sample Graphic Pictures	130

APPENDIX A

Sample Programs

Ø5 REM FAST RANDOM LINES

1Ø GR

15 X = INT(39*RND(1)) + 1

2Ø Y = INT(39*RND(1)) + 1

25 C = INT(15*RND(1)) + 1

3Ø COLOR=C

35 VLIN Ø, Y AT X

4Ø HLIN Ø, X AT Y

45 GOTO 15

Ø5 REM X-OUT

1Ø GR

15 COLOR=1

2Ø FOR X=Ø TO 39

25 PLOT X, X

3Ø PLOT 39 - X, X

35 NEXT X


```
05 REM CLOCK
10 REM USES A LOOP IN A LOOP
15 FOR X=1 TO 60
20 FOR T=1 TO 780
25 NEXT T
30 PRINT X
35 NEXT X
```

```
05 REM TAB DEMONSTRATION
10 FOR X=1 TO 20
15 PRINT TAB (X); X; " < ---- TAB POSITION"
20 NEXT X
25 PRINT "AND SO ON---- >"
```

Ø1 REM ***BIG LETTERS***
Ø5 REM SEE APPENDIX F
1Ø REM FOR MORE IDEAS
15 GR:COLOR=2
2Ø VLIN 16, 22 AT 8
25 VLIN 19, 22 AT 12
3Ø VLIN 19, 22 AT 15
35 VLIN 19, 22 AT 22
4Ø VLIN 19, 22 AT 26
45 VLIN 2Ø, 21 AT 3Ø
5Ø VLIN 2Ø, 21 AT 33
55 HLIN 8, 11 AT 19
6Ø HLIN 15, 18 AT 19
65 HLIN 15, 18 AT 2Ø
7Ø HLIN 15, 18 AT 22
75 HLIN 31, 32 AT 19
8Ø HLIN 31, 32 AT 22

```
01 REM ANIMATION DEMO
05 GR
10 FOR X=1 TO 39
15 COLOR=13
20 PLOT X, 20
30 COLOR=0
35 PLOT X - 1, 20
40 NEXT X
45 PLOT 39, 20
99 GOTO 10
```

```
05 REM STRING MANIPULATION
10 REM STUDY THE NEW WORDS
15 REM IN THE APPLE MANUAL
20 HOME
30 INPUT "WHAT'S YOUR NAME "; N$
35 FOR X=1 TO LEN (N$)
40 L$ = LEFT $ (N$,X)
50 PRINT L$
60 NEXT X
```

```
05 REM BLOCK ADDRESS TUTOR
10 HOME
15 PRINT "TYPE A BLOCK ADDRESS"
20 PRINT "IT MUST HAVE 2 NUMBERS"
25 PRINT "SEPARATED BY A COMMA"
30 PRINT "USE NUMBERS FROM 0 TO 39"
35 INPUT X, Y
40 GR
45 COLOR=12
50 PLOT X, Y
55 GOTO 35
```

```
05 REM HLIN TUTOR
10 HOME
15 PRINT TAB(11) " ***HLIN TUTOR*** "
20 PRINT:PRINT
25 PRINT " WHERE DO YOU WANT YOUR LINE? "
30 PRINT:PRINT
35 PRINT " TYPE A NUMBER "
40 PRINT:PRINT
45 PRINT " USE A NUMBER FROM 0 TO 39 "
50 INPUT A
55 GR
60 COLOR=3
65 HLIN 0, 39 AT A
70 GOTO 50
```

```
05 REM PUSH-PUSH PROGRAM
10 TEXT:HOME
15 FOR X=0 TO 243
20 PRINT TAB(X+1) "- - - - - > PUSH "
25 VTAB 1
30 FOR C = 1 TO 100:NEXT C
35 NEXT X
```

```
05 REM PUSH-PUSH PROGRAM
10 TEXT:HOME
15 FOR X=0 TO 243
20 PRINT TAB(X+1) "----- > PUSH"
25 VTAB 1
30 FOR C = 1 TO 100:NEXT C
35 NEXT X
```

```
05 REM LAZER GUN
10 GR:COLOR=1
15 INPUT " PUSH RETURN " ; A$
20 IF A$ < > " " THEN 30
25 GOTO 10
30 FOR X=0 TO 15
35 COLOR=1
40 PLOT X, 39 - X
50 PLOT 30 - X, 39 - X
60 NEXT X
70 GOTO 10
```

APPENDIX B


Saving Programs on Cassette Tape

When you turn the computer off, your program is erased. There are two ways to save your program. You can save programs on disks or cassettes. This section is about using a cassette recorder. The next section is about using a disk drive. You'll need a cassette tape recorder. You also need a special cord to hook it to the computer. If you have these things, read on.

Saving Programs on Cassette Tape

Follow these steps carefully.

1. Find someone who knows how to do this.
Ask that person to help you.
2. Get a **very good** new cassette tape. It should not have music or talk on it.
3. Be sure the recorder is plugged in right.
Study the picture on the next page.



4. Rewind the tape.
5. If you have a tape counter, set it to 0.
6. Push the PLAY button. Run the tape ahead to 10 on the counter.
7. Write down the number on the tape counter. You'll need it later to find your program. Also write down what your program does.
8. Type SAVE on the computer. **Do not** press RETURN yet.
9. Press down the PLAY and RECORD buttons. They should both stay down.
10. Press RETURN on the computer. The computer will beep when it starts to record. When it stops recording it will also beep. Then you'll see the cursor on the T.V.
11. Do all this over again. Leave a little room on the tape before your second SAVE.

Caution

When you're done, **push the stop button on the recorder.** This protects the recorder. Never leave a button down on a recorder. It can mess up the recorder.

Using Recorded Programs

To use a recorded program do this:

1. Put the tape into the recorder. Be sure the side you want is up.
2. Rewind the tape. Set the tape counter to zero.
3. Look at the tape counter. Run the tape ahead to the start of your program.
4. Rewind the tape just a little.
5. Type LOAD program on the computer.
6. Push **only** the PLAY button down.
7. Push the RETURN button on the computer. The T.V. will go blank. Wait quite a long time.
8. When done the computer will put the cursor on the screen. Now you can RUN the program.
9. If this does not work, **try again!**

Caution


When done, push the stop button on the recorder. This protects the recorder.

APPENDIX C

Saving Programs On Disks





Saving programs on disks is very easy. But, some disks should **not** have programs recorded on them.

1. Ask someone who knows which disk you should use.
2. Look at the next page. Study it carefully. Disks must be handled very carefully!
3. A disk has a "write protect" notch. Find this notch in the picture below. You **cannot** save a program when this notch is covered up. But, **don't uncover a "write protect" notch without asking someone.**



"write protect"
notch.

How to Handle Disks



4. Gently put the disk in the drive slot. Oval opening should go first. Label should be up. Then close the drive door.
5. Think up a name for your program.
6. Type `SAVE program name` and push RETURN.
7. You'll hear the disk spin for a few seconds. The computer will put the cursor on the T.V.

Getting Programs from Disk

1. Gently put the disk with your program on it into the drive. Close the door.
2. Type `LOAD program name` and push RETURN.
3. You'll hear the disk spin. Then the computer will put the cursor on the T.V.
4. Now, you can either `LIST` or `RUN` the program.


APPENDIX D

Graphics Screen Locations

FIRST COUNT OVER ➡

THEN COUNT DOWN ↓

START



APPENDIX E

Suggestions For Teachers & Parents

NOTE: TEACHING COMPUTER PROGRAMMING TO KIDS AND OTHER BEGINNERS, a notebook by the author, serves as a teacher's manual for the present volume. It has dozens of lessons, activities, and suggestions for teaching programming. In addition, it contains lessons specifically designed to introduce the present volume. Although not critical, it would be highly desirable for the reader to use this manual before and along with the present volume.

General Suggestions

1. If at all possible, get one copy of this book for each student. Let the kids mark in the book and put notes in the margins. If you can't get a copy for everyone, have every student keep a notebook of what they've learned.
2. Encourage students to read ahead in their free time. If they do, they won't waste as much time when they're at the computer.
3. For best results, each student needs two, one-half hour sessions at the computer each week. One session per week is a minimum. Most teachers pair up students. This is helpful because one can often see errors and understand words the other one might miss.
4. If you have some computer games, save them until later. Do **not** use them initially because most students would rather play games than learn to program. Later, you can **use the game tapes as rewards** for well written, original, student programs.

5. Get a chapter or two ahead of the kids. You don't have to be an excellent programmer to teach programming but you do have to stay a chapter or two ahead of the kids. It would also be helpful if you studied the Apple II manuals that came with the computer.
6. Build a cadre of expert kids. Let a few older students in your school or neighborhood quickly learn to program your computer. Then, get them to help you teach the younger kids to program. (Sometimes one kid can explain something to another better than an adult.)
7. Empathize with frustrated kids. Let's face it, it's hard to get a program to work exactly right the first time. Most programs must be "debugged." Kids aren't used to doing something over and over again until it's right. Be as understanding and helpful as you possibly can.
8. You'll constantly need to encourage kids to try writing their own programs. Doing what's in this manual (except for the Experiment sections) isn't programming. It's studying and typing. Eventually, kids will need to write their own **original** programs.
9. This manual uses graphics as a vehicle for motivating students to write their own programs. Make this as easy as possible for kids by making extra graphics layout sheets readily available. Hold computer **Art** classes and have a contest every month for the best computer graphic.
10. Very young students, grades K 2, who are poor readers might have trouble reading this manual. There are two ways you can eliminate this problem. First, you can pair up a good reader and a poor reader. Second, you can have

a parent volunteer or aid help the poor readers. If you do have an adult help, instruct them **not** to push the keys for the kids.

11. In pilot testing it was found that students weren't very interested in an approach that emphasized PRINT statements. They viewed PRINT lines as a bit of a gyp, e.g., you type 10 PRINT "HELLO" and the computer gives you one word instead of the 3 you typed. Hence, PRINT is emphasized less in this manual than some other manuals. You are certainly welcome to increase the emphasis on PRINT statements if you feel such an emphasis is beneficial.
12. Enjoy — this manual is intended to be pretty much self instructional. Sit back and watch the students learn — just help them over the hurdles.

Suggestions Regarding Hardware

1. This manual assumes that you have a color monitor attached to the Apple II. If you have a black and white monitor, the graphics programs in this manual will run but they will not run as described in the text. They will also not be as interesting to students.
2. Recently, Apple II computers have been coming from the factory with BASIC resident in ROM. That means they can be programmed in BASIC without disk drives or a cassette recorder connected. If you have disk drives, and a lot of beginning student programmers are going to use your computer, disconnect the drives. This will reduce turn-on time and hassles. During about the first half or more of this book the students have no need to save programs. (The programs are short enough that they can be easily entered by the students.)
3. Apple's Text versus Graphics mode can cause problems for beginning programmers, especially very young ones. Clearly describing this situation to the students, frequently reminding them of it, and placing a sign about it by the computer are about all that can be done. (It would sure be nice if in the future Apple computers would put Text and Home keys on their computers or at least configure them to default to the Text mode.)
4. In a classroom situation, it is better to instruct kids not to turn the computer off. This saves "wear and tear" on the computer and it also saves "turn on time." Simply have students type HOME and NEW to clear the screen before they begin programming.
5. If students are going to use a cassette tape recorder **you will need to** determine what volume level works and put a piece of tape over the volume control.

6. Static electricity can be a problem around computers. This is especially true if you have carpet on the floor and the indoor relative humidity is low. You might need to teach the kids to "ground themselves out" by touching a metal desk or similar object before touching the computer.
7. In general, the more you can teach kids about computer hardware the better. If they know how it works they can take better care of it. This applies to teaching such things as "disks are a magnetic medium," "key contacts can get dirty," etc. If you have very little knowledge in this area, invite someone who knows about these things to talk to your kids.

Specific Suggestions

Chapters 1-2

Show the students the computer. Point out the various parts (see chapter two). Discuss classroom procedures for signing-up to use the computer. Post and talk with the students about the rules for computer use.

Chapter 3

Before you allow students to use the computer tell them about the function of the **SHIFT** and **RETURN** keys. Be sure to tell them that the zero on the keyboard has a line through it (Ø). Younger students will usually spend several sessions at the computer just typing.

Chapter 4

Before students start on this chapter, explain the **TEXT** and **GRAPHICS** modes. Specifically, teach them that to go from **GRAPHICS** to **TEXT** they will need to type:

TEXT	RETURN
HOME	RETURN
LIST	RETURN

In this chapter students practice using the **PLOT** statement and typing. Block addresses are explained later.

Now would be a good time to start keeping a master list of **BASIC** words learned on the chalkboard or on a large poster. Also, put the **IMPORTANT BASIC WORDS** chart and the **COLOR NUMBER CHART** by the computer for quick student reference.

Chapter 5

This chapter teaches the use of an X,Y (over,down) coordinate system. Older students who have had graphing in math will breeze through this chapter. Younger students may need additional help from the teacher. Type in one of the **BLOCK ADDRESS TUTORIAL** programs (Appendix A) and let each student run the program several times. (Once you've typed and debugged the program save it on cassette or disk for future use.)

This chapter also introduces a very useful "shortcut," multiple statement lines, e.g., `TEXT:HOME:LIST`. For older students, explain this BASIC feature, but discourage them from using it in their programs. It makes reading programs much more difficult for them. For younger students, just let them use it as a "short cut" — explaining it might confuse them.

Chapter 6

The lesson in this chapter, Your Own Design, is very important. Students need to be continually encouraged to write original programs. Just doing what's in this book isn't programming, it's typing and studying. Whenever you see an **Experiment** section in this manual, let students put the book aside and write a few of their own programs. Now would be a good time for a graphic design contest.

Chapter 7

This should be a very rewarding chapter for students. `VLINE` and `HLINE` are very powerful BASIC statements. Younger students may, however, need some additional tutoring to firmly grasp the use of these statements. Use the terminology; `FROM TO DOWN`, or `FROM TO OVER`.

Chapter 8

Chapter eight introduces quite a few new concepts. Be certain students understand PRINT punctuation (commas and semicolons), GOTO and STRINGS before they move on to Chapter 9.

The ability to use the computer in calculator mode (just typing PRINT 2+2 without a line number) was purposefully not covered because it's easier just to use a hand held calculator. You're welcome to tell students of this feature of the computer.

Chapter 9

This is probably the most critical chapter in the book. Anyone who wants to become a proficient programmer must have a thorough understanding of do loops. (Some authors call these FOR-NEXT loops.) Take plenty of time to explain to anyone having difficulty how a do loop works. For younger students simulate a do loop by having a child role play a computer as it executes a program with a do loop in it. Keep the count on the chalkboard as the student does the task over and over again. For older students, you might want to discuss loops inside loops or " nested loops. "

Chapter 10

Very few new BASIC words are introduced in this chapter but the ability to read and understand how a program works becomes pretty significant in this chapter. Put several short programs on the chalkboard and teach students how to read and understand what the programs do without actually typing them into the computer.

Chapter 11

The main purpose of this chapter is to introduce a few new BASIC words. You might like to add additional BASIC words to those explained in this chapter.


Beyond This Book


Tentative plans are to publish in the near future a volume titled, **COMPUTER PROGRAMMING FOR KIDS AND OTHER BEGINNERS: BOOK TWO**. Until that volume is available, you can have students go ahead by either teaching them additional BASIC words and concepts yourself or by having them study Apple manuals.


APPENDIX F


Sample Graphic Pictures


130































POSTSCRIPT

Both the author and the publisher of this book are dedicated to providing the best possible materials for teaching kids and other beginners to program and otherwise use microcomputers. We solicit your comments and suggestions. Please let us know what works and we'll try to pass it on to others. Feel free to write either of us at the address below.

Yours truly,


Sterling Swift, Publisher


Royal Van Horn, Author

7901 South IH-35
Austin, Texas 78744

